
TITLE2'*** AMOSTRAGEM DE POPULAÇÕES DE ANIMAIS SILVESTRES ***'; 
TITLE4'*** ESTUDO DE AMOSTRAGEM ESTRATIFICADA  ***'; 
DATA A; 
INPUT ESTRATO NTOTH NESTH VARAMOS MEDIA; 
*NTOTH = NÚMERO TOTAL DE UNIDADES AMOSTRAIS NO ESTRATO H 
 NESTH = NÚMERO DE UNIDADES AMOSTRAIS TOMADAS NO ESTRATO H 
 VARAMOS = VARIÂNCIA DA AMOSTRA DE CADA ESTRATO 
 MEDIA = MÉDIA DE CADA ESTRATO; 
DUM=1; 
 DATALINES; 
 1 200 10 400 15 
 2 100 10 900 10 
 3 400 10 400 22 
 ; 
 PROC SORT DATA=A; 
      BY DUM; 
PROC MEANS DATA=A NOPRINT; 
      BY DUM; 
VAR NTOTH NESTH; 
OUTPUT OUT=B SUM=NTOT NAMO; 
DATA C; 
     MERGE A B; 
     BY DUM; 
CAL1=NTOTH*MEDIA/NTOT; 
CAL2=(NTOTH/NTOT)**2; 
WH=NESTH/NAMO; 
CAL3=(NTOTH**2)*VARAMOS/WH; 
CAL4=NTOTH*VARAMOS; 
CORRE=1-(NESTH/NTOTH);*CORREÇÃO PARA POPULAÇÃO FINITA; 
VARAMOEST=(VARAMOS/NESTH)*CORRE; 
TOTALH=NTOTH*MEDIA;* TOTAL DO ESTRATO H; 
VARTOTH=(NTOTH**2)*VARAMOEST;*VARIÂNCIA DA AMOSTRA DO ESTRATO H; 
PROC MEANS DATA=C NOPRINT; 
    VAR CAL1 TOTALH VARTOTH CAL3 CAL4 NESTH NTOTH; 
OUTPUT OUT=D SUM=; 
DATA E; 
    SET D; 
T=TINV(.975,NESTH-1); *VALOR DE T PARA 95% PROBABILIDADE; 
MEDEST=CAL1; 
ED=10;*ERRO DESEJADO 10%; 
TOTAL=TOTALH; 
ICTOTAL=T*SQRT(VARTOTH); 
ERRO=(ICTOTAL/TOTAL)*100; 
NAST=CAL3/(CAL4+(((NTOTH**2)*(MEDEST**2)*(ED**2))/((100**2)*(T**2)))); 
PROC PRINT DATA=E LABEL SPLIT='*' NOOBS; 
VAR MEDEST TOTAL ICTOTAL ERRO NAST; 
LABEL MEDEST='MÉDIA DA*AMOSTRAGEM*ESTRATIFICADA' 
      TOTAL='TOTAL' 
   ICTOTAL='INTERVALO DE*CONFIANÇA*DO TOTAL' 
      ERRO='ERRO*AMOSTRAL*(%)' 
      NAST='INTENSIDADE*AMOSTRAL'; 
FORMAT MEDEST ERRO COMMAX8.2 ICTOTAL TOTAL NAST 8.0;RUN; 
 


